# $Id: PKGBUILD 151414 2012-02-26 10:40:20Z pierre $
#Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=libxevie
pkgver=1.0.3
pkgrel=2
pkgdesc="X11 EvIE extension library"
arch=('i686' 'x86_64')
url="http://xorg.freedesktop.org/"
license=('custom')
depends=('libxext')
makedepends=('xorg-util-macros' 'evieext')
options=('!libtool')
source=("${url}/releases/individual/lib/libXevie-${pkgver}.tar.bz2")
sha1sums=('84e2dc6e35c4a82f1a4d1fe4723fba58364c3bec')

build() {
  cd "${srcdir}/libXevie-${pkgver}"
  ./configure --prefix=/usr --sysconfdir=/etc --disable-static
  make
}

package() {
  cd "${srcdir}/libXevie-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -m755 -d "${pkgdir}/usr/share/licenses/${pkgname}"
  install -m644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/"
}
